
1 INTRODUCTION. Rabinowitz and Wagon [8] present a “remarkable” algorithm for com-
puting the decimal digits ofπ, based on the expansion

π =
∞

∑
i=0

(i!)22i+1

(2i +1)!
. (1)

Their algorithm uses only bounded integer arithmetic, and is surprisingly efficient. Moreover, it admits
extremely concise implementations. Witness, for example, the following (deliberately obfuscated) C pro-
gram due to Dik Winter and Achim Flammenkamp [1, p. 37], which produces the first 15,000 decimal
digits of π:

a[52514],b,c=52514,d,e,f=1e4,g,h;main(){for(;b=c-=14;h=printf("%04d",
e+d/f))for(e=d%=f;g=--b*2;d/=g)d=d*b+f*(h?a[b]:f/5),a[b]=d%--g;}

Rabinowitz and Wagon call their algorithm aspigot algorithm, because it yields digits incrementally
and does not reuse digits after they have been computed. The digits drip out one by one, as if from a leaky
tap. In contrast, most algorithms for computing the digits ofπ execute inscrutably, delivering no output
until the whole computation is completed.

However, the Rabinowitz–Wagon algorithm has its weaknesses. In particular, the computation is in-
herently bounded: one has to commit in advance to computing a certain number of digits. Based on this
commitment, the computation proceeds on an appropriate finite prefix of the infinite series (1). In fact, it
is essentially impossible to determine in advance how big that finite prefix should be for a given number
of digits—specifically, a computation that terminates with nines for the last few digits of the output is
inconclusive, because there may be a “carry” from the first few truncated terms. Rabinowitz and Wagon
suggest that “in practice, one might ask for, say, six extra digits, reducing the odds of this problem to one
in a million” [8, p. 197], a not entirely satisfactory recommendation. Indeed, the implementation printed
at the end of their paper is not quite right [1, p. 82], sometimes printing an incorrect last digit because the
finite approximation of the infinite series is one term too short.

We propose a different algorithm, based on the same series (1) forπ but avoiding these problems.
We also show the same technique applied to other characterizations ofπ. No commitment need be made
in advance to the number of digits to be computed; given enough memory, the programs will generate
digits ad infinitum. Once more (necessarily, in fact, given the previous property), the programs are spigot
algorithms in Rabinowitz and Wagon’s sense: they yield digits incrementally and do not reuse them after
producing them. Of course, no algorithm using a bounded amount of memory can generate a nonrepeat-
ing sequence such as the digits ofπ indefinitely, so we have to allow arbitrary-precision arithmetic, or
some other manifestation of dynamic memory allocation. Like Rabinowitz and Wagon’s algorithm, our
proposals are not competitive with state-of-the-artarithmetic-geometric meanalgorithms for computingπ

[2], [9]. Nevertheless, our algorithms are simple to understand and admit almost as concise an implemen-
tation. As evidence to support the second claim, here is a (deliberately obscure) program that will generate
as many digits ofπ as memory will allow:

> pi = g(1,0,1,1,3,3) where
> g(q,r,t,k,n,l) = if 4*q+r-t<n*t
> then n : g(10*q,10*(r-n*t),t,k,div(10*(3*q+r))t-10*n,l)
> else g(q*k,(2*q+r)*l,t*l,k+1,div(q*(7*k+2)+r*l)(t*l),l+2)

The remainder of this paper provides a justification for the foregoing program, and some others like it.
These algorithms exhibit a pattern that we callstreaming[3]. Informally, a streaming algorithm con-

sumes a (potentially infinite) sequence of inputs and generates a (possibly infinite) sequence of outputs,
maintaining some state as it goes. Based on the current state, at each step there is a choice between
producing an element of the output and consuming an element of the input. Streaming seems to be a com-
mon pattern for various kinds ofrepresentation changers, including several data compression and number
conversion algorithms.

The program under discussion is written in Haskell [5], a lazy functional programming language.
As a secondary point of this paper, we hope to convince the reader that such languages are excellent
vehicles for expressing mathematical computations, certainly when compared with other general-purpose
programming languages such as Java, C, and Pascal, and arguably even when compared with computer
algebra systems such as Mathematica. In particular, a lazy language allows direct computations with
infinite data structures, which require some kind of indirect representation in most other languages. The
Haskell program presented earlier has been compressed to compete with the C program for conciseness, so
we do not argue that this particular one is easy to follow—but we do claim that the later Haskell programs
are.

2 LAZY FUNCTIONAL PROGRAMMING IN HASKELL. To aid the reader’s
understanding, we start with a brief (and necessarily incomplete) description of the concepts of functional
programming (henceforth FP) and laziness and their manifestation in Haskell [5], the de facto standard
lazy FP language. Further resources, including pointers to tutorials and free implementations for many
platforms, can be found at the Haskell website [6].

FP is programming withexpressionsrather thanstatements. Everything is a value, and there are no
assignments or other state-changing commands. Therefore, a pure FP language isreferentially transparent:
an expression may always be substituted for one with an equal value, without changing the meaning of the
surrounding context. This makes reasoning in FP languages just like reasoning in high-school algebra.

Here is a simple Haskell program:

> square :: Integer -> Integer
> square x = x * x

Program text is marked with a “>” in the left-hand column, and comments are unmarked. This is a simple
form of literate programming, in which the emphasis is placed on making the program easy for people
rather than computers to read. Code and documentation are freely interspersed; indeed, the manuscript for
this article is simultaneously an executable Haskell program.

The first line in thesquare program is a type declaration; the symbol “:: ” should be read “has
type.” Thus,square has typeInteger -> Integer , that is, it is a function fromInteger s to
Integer s. (Type declarations in Haskell are nearly always optional, because they can be inferred, but we
often specify them anyway for clarity.) The second line gives a definition, as an equation: in any context,
a subexpression of the formsquare x for anyx may be replaced safely byx * x .

Lists are central to FP. Haskell uses square brackets for lists, so[1,2,3] has three elements, and[]
is the empty list. The operator “: ” prefixes an element; so[1,2,3] = 1 : (2 : (3 : [])) . For
any typea, there is a corresponding type[a] of lists with elements drawn from typea, so, for exam-
ple, [1,2,3] :: [Integer] . Haskell has a very convenientlist comprehensionnotation, analogous
to set comprehensions; for instance,[ square x | x <- [1,2,3] ] denotes the list of squares
[1,4,9] .

For the “substitution of equals for equals” property to be universally valid, it is important not to evaluate
expressions unless their values are needed. For example, consider the following Haskell program:

> three :: Integer -> Integer
> three x = 3

> nonsense :: Integer
> nonsense = 1 + nonsense

The first definition is of a function that ignores its argumentx and always returns the integer3; the second
is of a value of typeInteger , but one whose evaluation never terminates. For substitutivity to hold,
and in particular forthree nonsense to evaluate to3 as the equation suggests, it is important not to
evaluate the function argumentnonsense in the function applicationthree nonsense . With lazy
evaluation, in which evaluation is demand-driven, no expression is evaluated unless and until its value is
needed to make further progress.

A useful by-product of lazy evaluation is the ability to handle infinite data structures: they are evaluated
only as far as is necessary. We illustrate this with a definition of the infinite sequence of Fibonacci numbers:

> fibs :: [Integer]
> fibs = f (0,1) where f (a,b) = a : f (b,a+b)

(Here,(0,1) is a pair ofInteger s, and thewhere clause introduces a local definition.) Evaluating
fibs never terminates, of course, but computing a finite prefix of it does. For example, with

> take :: Integer -> [Integer] -> [Integer]
> take 0 xs = []
> take (n+1) [] = []
> take (n+1) (x:xs) = x : take n xs

(so take takes two arguments, anInteger and a[Integer] , and returns another[Integer] ), we
havetake 10 fibs = [0,1,1,2,3,5,8,13,21,34] , which terminates normally.

Functions may bepolymorphic, defined for arbitrary types. Thus, the functiontake in fact works for
lists of any element type, since elements are merely copied and not further analyzed. The most general
type assignable totake is Integer -> [a] -> [a] for an arbitrary typea.

Because this is FP, functions are “first-class citizens” of the language, with all the rights of any other
type. Among other things, they may be passed as arguments to and returned as results fromhigher-order
functions. For example, with the definition

> map :: (a->b) -> [a] -> [b]
> map f [] = []
> map f (x:xs) = f x : map f xs

the list comprehension[ square x | x <- [1,2,3] ] is equal tomap square [1,2,3] .
Note thatmap and take are curried. In fact, the function type former-> associates to the right, so
the type ofmap is equivalent to(a->b) -> ([a]->[b]) , which one might read as saying thatmap
takes a function of typea->b and transforms it into one of type[a]->[b] .

3 RABINOWITZ AND WAGON’S SPIGOT ALGORITHM. Rabinowitz and
Wagon’s algorithm is based on the series (1) forπ, which expands out to the expression

π = 2+
1
3

(
2+

2
5

(
2+

3
7

(
· · ·

(
2+

i
2i +1

(
· · ·

)))))
. (2)

This expression forπ can be derived from the well-known Leibniz series

π
4

=
∞

∑
i=0

(−1)i

2i +1
(3)

using Euler’s convergence-accelerating transform, among several other methods [7]. One can view expres-
sion (2) as representing a number(2;2,2,2, . . .) in a mixed-radix baseB = (1

3, 2
5, 3

7, . . .), in the same way
that the usual decimal expansion

π = 3+
1
10

(
1+

1
10

(
4+

1
10

(
1+

1
10

(
5+ · · ·

))))
represents(3;1,4,1,5, . . .) in the fixed-radix baseF10, whereFm = ( 1

m, 1
m, 1

m, . . .). The task of computing
the decimal digits ofπ is then simply a matter of converting from baseB to baseF10.

We considerregular representations. For a regular representation in decimal, every digit after the
decimal point is in the range[0,9]. The decimal number(0;9,9,9, . . .) with a zero before the point and
maximal digits afterwards represents 1. Accordingly, regular decimal representations with a zero before the
point lie between 0 and 1. By analogy, for a regular representation in baseB, the digit in positioni after the
point (the first after the point being in position 1) is in the range[0,2i]. The number(0;2,4,6, . . .) with zero
before the point and maximal digits afterwards represents 2. Accordingly, regular baseB representations
with zero before the point lie between 0 and 2. (We call the representations “regular” rather than “normal”
because they are not unique.)

Conversion from baseB to decimal proceeds as one might expect. The integer part of the input
becomes the integer part of the output. The fractional part of the input is multiplied by ten; the integer
part of this becomes the first output digit after the decimal point, and the fractional part is retained. This is
again multiplied by ten; the integer part of this becomes the second output digit after the point; and so on.

Multiplying a number in baseB by ten is achieved simply by multiplying each digit by ten. However,
that yields an irregular result, because some of the resulting digits may be too big. Regularization proceeds
from right to left, reducing each digit as necessary and propagating any carry leftwards.

The only remaining problem is in computing the integer part of a number(a0;a1,a2,a3) in baseB. This
is eithera0 or a0 +1, depending on whether the remainder(0;a1,a2,a3) is in [0,1) or [1,2). (In principle,
the remainder could equal 2; but, in practice, this cannot happen in the computation of an irrational such as
π.) Therefore, Rabinowitz and Wagon’s algorithm temporarily buffers any nines that are produced, until it
is clear whether or not there could be a carry that would invalidate them.

This whole conversion is performed on afinite number(2;2,2,2, . . . ,2) in baseB—necessarily, as
regularization proceeds from right to left. Rabinowitz and Wagon provide a bound on the number of baseB
digits needed to yield a given number of decimal digits. In fact, as mentioned earlier, they underestimate by
one in some cases:b10n/3c digits is usually sufficient, but sometimesb10n/3c+1 input digits is necessary
(and sufficient) forn decimal digits. (Here, “bxc” denotes the greatest integer not larger thanx.) Again, as
noted, this does not mean that thosen decimal digits are all correct digits ofπ, only that then-digit decimal
number produced is within 5×10−n of the desired result.

4 STREAMING ALGORITHMS. We turn now to streaming algorithms, by way of a
simpler example than computing the digits ofπ. Consider the problem of converting a fraction in the
interval [0,1] from one base to another. We represent fractions as digit sequences, and for simplicity (but
without loss of generality) we consider only infinite sequences. For this reason, we cannot consume all
the input before producing any output; we must alternate between consumption and production. The com-
putation will therefore maintain some state depending on the inputs consumed and the outputs produced
thus far. Based on that state, it will either produce another output, if that is possible given the available
information, or consume another input if it is not. This pattern is captured by the following higher-order
function:

> stream :: (b->c) -> (b->c->Bool) -> (b->c->b) -> (b->a->b) ->
> b -> [a] -> [c]
> stream next safe prod cons z (x:xs)
> = if safe z y
> then y : stream next safe prod cons (prod z y) (x:xs)
> else stream next safe prod cons (cons z x) xs
> where y = next z

This defines a functionstream taking six arguments. The result of applyingstream is an infinite list
of output terms, each of typec . The last argumentx:xs is a list of input terms, each of typea; the
first element or “head” isx , and the infinite remainder or “tail” isxs . The penultimate argumentz is the
state, of typeb. The other four arguments (next of typeb->c , safe of typeb->c->Bool , prod of
type b->c->b andcons of type b->a->b ) are all functions. From the statez the function produces
a provisional output termy = next z of type c . If y is safe to commit to from the current statez
(whatever input terms may come next), then it is produced, and the state adjusted accordingly usingprod ;
otherwise, the next termx of the input iscons umed into the state. This process continues indefinitely:
the input is assumed never to run out, and, if the process is productive, the output never terminates.

In the case of conversion from an infinite digit sequence in baseFm to an infinite sequence in baseFn,
clearly both the input and output elements are of typeInteger . The state maintained is a pair(u,v) of
Rational s, satisfying the “invariant” (that is, a property that is established before a loop commences,
and is maintained by each iteration of that loop) that the original input

1
m

(
x0 +

1
m

(
x1 + · · ·

))
is equal to

1
n

(
y0 +

1
n

(
y1 + · · ·+ 1

n

(
y j−1 +v×

(
u+

1
m

(
xi +

1
m

(
xi+1 + · · ·

))))))
wheni input termsx0,x1, . . . ,xi−1 have been consumed andj output termsy0,y1, . . . ,y j−1 have been pro-
duced. Initiallyi and j are zero, so the invariant is established withu = 0 andv = 1. In order to maintain
the invariant, the state(u,v) should be transformed to(u−y/(nv),nv) when producing an additional out-
put termy and to(x+um,v/m) when consuming an additional input termx. The value of the remaining
input

1
m

(
xi +

1
m

(
xi+1 + · · ·

))
ranges between 0 and 1, so the next output term is determined provided thatnvuandnv(u+1) have the
same integer part orfloor . This justifies the following streaming algorithm:

> convert :: (Integer,Integer) -> [Integer] -> [Integer]
> convert (m,n) xs = stream next safe prod cons init xs
> where
> init = (0%1, 1%1)
> next (u,v) = floor (u*v*n’)
> safe (u,v) y = (y == floor ((u+1)*v*n’))
> prod (u,v) y = (u - fromInteger y/(v*n’), v*n’)
> cons (u,v) x = (fromInteger x + u*m’, v/m’)
> (m’,n’) = (fromInteger m, fromInteger n)

(Here, “%” constructs aRational from two Integer s, “==” is the comparison operator, and the func-
tion fromInteger coercesInteger s toRational s.)

For example, 1/e is 0.1002210112. . . in base 3 and 0.2401164352. . . in base 7. Therefore, applying the
functionconvert (3,7) to the infinite list[1,0,0,2,2,1,0,1,1,2... should yield the infinite
list [2,4,0,1,1,6,4,3,5,2... . The first few states through which execution of this conversion
proceeds are illustrated in the following table:

input 1 0 0 2 2 1

state 0
1, 1

1
1
1, 1

3
3
1, 1

9
9
1, 1

27
9
7, 7

27
41
7 , 7

81
137
7 , 7

243
418
7 , 7

729
10
49,

49
729

10
49,

343
729 · · ·

output 2 4 0

The middle row shows consecutive values of the state(u,v). The upper row shows input digits consumed,
above the state resulting from their consumption. The lower row shows output digits produced, below the
state resulting from their production. Notice that outputs are produced precisely when the corresponding
state(u,v), which is in the previous column, is safe; that is, whenb7uvc = b7(u+1)vc. For example, the
fourth state(9

1, 1
27) is safe, becauseb7× 9

1 ×
1
27c = 2 = b7× (9

1 +1)× 1
27c.

This paper is not the place to make a more formal justification for the correctness of this program,
although it is not hard to establish from the invariant stated. Nevertheless, it is possible toderive the
streaming program from a specification expressed in terms of independent operations for expanding and
collapsing digit sequences, using a general theory of such algorithms [3]. (In the general case, either the
input or the output or both may be finite. We have stuck to the simple case of necessarily-infinite lists
here, because that is all that is needed for computing the digits ofπ.) We have found this pattern of
computation in numerous problems concerningchanges of data representation, of which conversions be-
tween number formats are a representative example. Consequently, we have been calling such algorithms
metamorphisms.

5 A STREAMING ALGORITHM FOR THE DIGITS OF π. The main prob-
lem with Rabinowitz and Wagon’s spigot algorithm is that it is bounded: one must make a commitment
in advance to the number of terms of the series (1) to use. This commitment arises because the process of
regularizing a number in baseB proceeds from right to left, hence works only for finite numbers in that
base.

It turns out that there is a rather simplestreaming algorithmfor regularizing infinite numbers in baseB.
This means that we can make Rabinowitz and Wagon’s algorithm unbounded: there is no longer any need
to make a prior commitment to a particular finite prefix of the expansion ofπ. However, we will not
say any more about this approach, for there is a more direct way of computing the digits ofπ from the
expression (2), to which we now turn.

One can view the expansion (2) as the composition

π =
(

2+
1
3
×

)(
2+

2
5
×

)(
2+

3
7
×

)
· · ·

(
2+

i
2i +1

×
)
· · · (4)

of an infinite series oflinear fractional transformationsor Möbius transformations. These are functions
taking x to qx+r

sx+t for integersq, r,s andt with qt− r s 6= 0—that is, yielding a ratio of integer-coefficient
linear transformations ofx. Such a transformation can be represented by the four coefficientsq, r,s andt,
and if they are arranged as a matrix

(q
s

r
t

)
then function composition corresponds to matrix multiplication.

> type LFT = (Integer, Integer, Integer, Integer)

> extr :: LFT -> Integer -> Rational
> extr (q,r,s,t) x = ((fromInteger q) * x + (fromInteger r)) /
> ((fromInteger s) * x + (fromInteger t))
> unit :: LFT
> unit = (1,0,0,1)
> comp :: LFT -> LFT -> LFT
> comp (q,r,s,t) (u,v,w,x) = (q*u+r*w,q*v+r*x,s*u+t*w,s*v+t*x)

(The first line introduces the abbreviationLFT for the type of four-tuples ofInteger s.)
The infinite composition of transformations in (4) converges, in the following sense. Although the

products of finite prefixes of the composition have coefficients that grow without bound, the transforma-
tions represented by these products map the interval[3,4] onto converging subintervals of itself. (This is
easy to see, as each term is a monotonic transformation, reduces the width of an interval by at least a factor
of two, and maps[3,4] onto a subinterval of itself. Indeed, the same also holds for any tail of the infinite
composition.)

Therefore, equation (4) can be thought of as the representation of some real number, and computing
the decimal digits of this number is a change of representation, effectable by a streaming algorithm. The
streaming process maintains as its state an additional linear fractional transformation, representing the
required function from the inputs yet to be consumed to the outputs yet to be produced. This state is
initially the identity matrix; consumption of another input term is matrix multiplication; production of a
digit n is multiplication by

(10
0

−10n
1

)
, the inverse of the linear fractional transformation takingx to n+ x

10.
If the current state is the transformationz, then the next digit to be produced lies somewhere in the image
underz of the interval[3,4]; if the two endpoints of this image have the same integer part, then that next
digit is completely determined and it is safe to commit to it.

> pi = stream next safe prod cons init lfts where
> init = unit
> lfts = [(k, 4*k+2, 0, 2*k+1) | k<-[1..]]
> next z = floor (extr z 3)
> safe z n = (n == floor (extr z 4))
> prod z n = comp (10, -10*n, 0, 1) z
> cons z z’ = comp z z’

The definition oflfts uses a list comprehension, with generator[1..] , the infinite list ofInteger s
from 1 upwards. The list consists of the expression(k, 4*k+2, 0, 2*k+1) to the left of the vertical
bar, evaluated for each value ofk from 1 upwards. The first few terms are[(

1 6
0 3

)
,

(
2 10
0 5

)
,

(
3 14
0 7

)
, . . .

]
.

For example, the first term
(1

0
6
3

)
represents the transformation takingx to 1×x+6

0×x+3, or 2+ 1
3x.

The condensed program shown in section 1 can be obtained from this program by making various
simple optimizations. These include: unfolding intermediate definitions; exploiting the invariant that the

bottom left elements of every linear fractional transformation
(q

s
r
t

)
be 0; constructing the input trans-

formations in place; representing the sequence of remaining transformations simply by the indexk ; and
simplifying away one of the divisions.

Another optimization that can be performed is the elimination of any factors common to all four entries
resulting from a matrix multiplication. This optimization is valid, since linear fractional transformations
are invariant under scaling of the matrix, and helpful, as it keeps the numbers small. (In fact, it is better
still to perform this cancellation less frequently than every iteration.)

6 MORE STREAMING ALGORITHMS FOR THE DIGITS OF π. The
expression (2) turns out not to be a very efficient one for computation. Each term shrinks the range by a
factor of about a half, so more than three terms are required on average for every digit of the output. Better
sequences are known; the bookπ Unleashed[1] presents many. We conclude this paper with two more
applications of the streaming technique from section 4 to computingπ, using the same approach but based
on two of these different expressions.

Lambert’s expression. Here is a more efficient expression forπ, due to Lambert in 1770 [1, eq. (16.99)],
that yields two decimal digits for every three terms:

π =
4

1+
12

3+
22

5+
32

7+ · · ·

. (5)

Again, one can view this as an infinite composition of linear fractional transformations:

π =
(

4÷
)(

1+12÷
)(

3+22÷
)(

5+32÷
)
· · ·

(
2i−1+ i2÷

)
· · · .

After consumingi terms of the input, the remaining terms represent the composition(
2i−1+ i2÷

)(
2i +1+(i +1)2÷

)(
2i +3+(i +2)2÷

)
· · · ,

which denotes a value in the range[2i − 1,2i − 1+ i
2]. As before, we subject this infinite sequence to

a streaming process. This time, however, we maintain a state consisting of not just a linear fractional
transformation

(q
s

r
t

)
, but also the numberi of terms consumed thus far (needed in order to determine the

next digit to produce, which lies between
(q

s
r
t

)
(2i−1) and

(q
s

r
t

)
(2i−1+ i

2)).
This reasoning justifies the following program:

> piL = stream next safe prod cons init lfts where
> init = ((0,4,1,0), 1)
> lfts = [(2*i-1, i*i, 1, 0) | i<-[1..]]
> next ((q,r,s,t),i) = floor ((q*x+r) % (s*x+t)) where x=2*i-1
> safe ((q,r,s,t),i) n = (n == floor ((q*x+2*r) % (s*x+2*t)))
> where x=5*i-2
> prod (z,i) n = (comp (10, -10*n, 0, 1) z, i)
> cons (z,i) z’ = (comp z z’, i+1)

Gosper’s series. An even more efficient series forπ, yielding more than one decimal digit for each
term, is due to Gosper [4]:

π = 3+
1×1

3×4×5
×

(
8+

2×3
3×7×8

×
(
· · ·5i−2+

i(2i−1)
3(3i +1)(3i +2)

×·· ·
))

. (6)

Once more, we can view this as an infinite composition of linear fractional transformations, namely,

π =
(

3+
1×1

3×4×5
×

)(
8+

2×3
3×7×8

×
)
· · ·

(
5i−2+

i(2i−1)
3(3i +1)(3i +2)

×
)
· · · .

It is not hard to show that, after consumingi−1 terms of the input, the remaining terms denote a value in
the range[27

5 i− 12
5 , 27

5 i− 2333

53 ], which gives rise to the following program:

> piG = stream next safe prod cons init lfts where
> init = ((1,0,0,1), 1)
> lfts = [let j = 3*(3*i+1)*(3*i+2)
> in (i*(2*i-1),j*(5*i-2),0,j) | i<-[1..]]
> next ((q,r,s,t),i) = div (q*x+5*r) (s*x+5*t) where x = 27*i+15
> safe ((q,r,s,t),i) n = (n == div (q*x+125*r) (s*x+125*t))
> where x=675*i-216
> prod (z,i) n = (comp (10, -10*n, 0, 1) z, i)
> cons (z,i) z’ = (comp z z’, i+1)

(The let here is another form of local definition.)

A challenge. Gosper’s series (6) yields more than one digit per term on average, since the scaling
factors approach2

27 (which is less than1
10) from below. This suggests that we could dispense with the

test altogether and strictly alternate between consumption and production, as expressed by the following
conjecture. Eliminating the test would speed up the algorithm considerably.

Conjecture 1. Define the following functions:

n(z, i) =
⌊
z(27i+15

5 )
⌋

p((z, i),n) = (z
(10

0
−10n

1

)
, i)

c((z, i),z′) = (zz′, i +1)
s((z, i),n) = (n =

⌊
z(675i−216

125 )
⌋
) .

For i = 1,2. . ., let
xi =

( i(2i−1)
0

3(3i+1)(3i+2)(5i−2)
3(3i+1)(3i+2)

)
u0 = (

(1
0

0
1

)
,1)

ui = p(vi ,yi)
vi = c(ui−1,xi)
yi = n(vi) .

Then s(vi ,yi) holds for all i.

If this conjecture holds, then, in the following program, every value taken by the variablev satisfies the
conditionsafe v (next v) .

9



> piG2 = process next prod cons init lfts
> process next prod cons u (x:xs)
> = y : process next prod cons (prod v y) xs
> where v = cons u x
> y = next v

(Here,next , prod , cons , init , andlfts , as well as the predicatesafe mentioned in the claim, are
as in section 6.)

We have not been able to prove Conjecture 1, although we have verified it for the first thousand terms.
Perhaps some diligent reader can provide enlightenment. If it is valid, thenpiG2 does indeed produce
the digits ofπ, and the optimizations outlined at the end of section 5 can be applied topiG2 , yielding the
following program:

> piG3 = g(1,180,60,2) where
> g(q,r,t,i) = let (u,y)=(3*(3*i+1)*(3*i+2),div(q*(27*i-12)+5*r)(5*t))
> in y : g(10*q*i*(2*i-1),10*u*(q*(5*i-2)+r-y*t),t*u,i+1)

This is of comparable length to the compressed program given in section 1, but approximately five times
faster.

BRIEF DESCRIPTIVE SUMMARY. Rabinowitz and Wagon (in the April 1995 issue of
the MONTHLY) present aspigot algorithmfor computing the digits ofπ. A spigot algorithm yields its
outputs incrementally, and does not reuse them after producing them. Rabinowitz and Wagon’s algorithm
is inherentlybounded; it requires a commitment in advance to the number of digits to be computed. We
propose somestreaming algorithmsbased on the same and some similar characterizations ofπ, with the
same incremental characteristics, but without requiring the prior bound.
jeremy.gibbons@comlab.ox.ac.uk

